

# Index notation in Lean 4

Joseph Tooby-Smith  
*Reykjavik University, Reykjavik, Iceland*

November 5, 2024

## Abstract

The physics community relies on index notation to effectively manipulate tensors of specific types. This paper introduces the first formally verified implementation of index notation in the interactive theorem prover Lean 4. By integrating index notation into Lean, we bridge the gap between traditional physics notation and formal verification tools, making it more accessible for physicists to write and prove results within Lean. In the background, our implementation leverages a novel application of category theory.

The file you are currently reading is a draft. There will be typos, mistakes, sentences that don't make sense, notation that is not defined etc.

## 1. INTRODUCTION

In previous work [1], the author initiated the digitalization (or formalization) of high energy physics results using the interactive theorem prover Lean 4 in a project called HepLean. Lean is a programming language with syntax resembling traditional pen-and-paper mathematics. Users can write definitions, theorems, and proofs in Lean, which are then automatically checked for correctness using dependent-type theory. The HepLean project is driven by four primary motivations: to facilitate easier look-up of results through a linear storage of information; to support the creation and proof of new results using automated tactics and AI tools; to facilitate correctness checking of result in high energy physics; and to introduce new pedagogical methods for high-energy physics and computer science.

HepLean is part of a broader movement of projects to formalize parts, or all, of mathematics and science. The largest of these projects is Mathlib [2], which aims to formalize mathematics. Indeed, HepLean uses many results from Mathlib, and is built downstream thereof. Other projects in mathematics include the ongoing effort led by Kevin Buzzard to formalize the proof of Fermat's Last Theorem into Lean [3]. In the realm of the sciences, the paper [4] looks at absorption theory, thermodynamics, and kinematics in Lean, whilst the package SciLean [5], is a push in the direction of scientific computing within Lean.

Physicists rely heavily on specialized notation to express mathematical concepts succinctly. Among these index notation is particularly prevalent, as it provides a compact and readable way to represent specific types of tensors and operations between them. Such tensors form a backbone of modern physics.

Having a way to use index notation in Lean is crucial for the digitalisation of results from high energy physics. As well as, making results from high energy physics easier to write and prove in Lean, it will make the syntax more familiar to high energy physicists. Such an implementation is the subject of this paper. It is a challenge because, we need an implementation that is nice and flexible to use, but which is also formally well-defined and rigorous. We hope that the implementation presented here will



Figure 1: Overview of the implementation of index notation in Lean. The solid lines represent formally verified parts of the implementation.

not only enhances usability of Lean but also promotes the adoption of formal methods in the physics community.

As a taster of what is to come, in our implementation the result regarding Pauli matrices that  $\sigma^{\nu\alpha\beta}\sigma_{\nu}\alpha'\dot{\beta}' = 2\epsilon^{\alpha\alpha'}\epsilon^{\beta\beta'}$  is written in Lean as follows

```
{pauliCo | v α β ⊗ pauliContr | v α' β' = 2 · εL | α α' ⊗ εR | β β'}T
```

Lean will correctly interpret this result as a tensor expression with the correct contraction of indices and permutation of indices between the sides of the expression.

Previous implementations of index notation have been made in programming languages like Haskell [6]. However, the programs they appear in do not provide the formal verification capabilities inherent in Lean. The formal verification requirement of Lean introduces unique challenges in implementing index notation, necessitating (what we believe is) a novel solution.

In Section 2 of this paper, we will discuss the details of our implementation. In Section 3 we will give two examples of theorems and proofs using index notation. The first of these examples involves a lemma regarding the contraction of indices of symmetric and anti-symmetric tensors. The second involves examples related to the Pauli matrices and bi-spinors. We finish this paper in Section 4 by discussing potential future work related to this project.

## 2. IMPLEMENTATION OF INDEX NOTATION INTO LEAN 4

Our implementation of index notation can be thought of as three different representations of tensor expressions and maps between them. This is illustrated in Figure 1.

The first representation is *syntax*. This can (roughly) be thought of as the informal string that represents the tensor expression. It is what the user interacts with when writing results in Lean, and what appears in raw Lean files. The code snippet above, for Pauli matrices is an example of syntax.

The second representation of a tensor expression is a *tensor tree*. This representation is mathematically formal, but easy to use and manipulate. It is a structured tree which has different types of nodes for each of the main operations that one can perform on tensors.

The process of going from syntax to a tensor tree is done via an elaborator which follows a number of (informally defined) rules.

The third and final representation is a bona-fide *tensor*. This representation is the mathematical object that we are actually interested in. However, in going to this representation we lose the structure of the tensor expression itself, making it somewhat difficult to work with.

The process of going from a tensor tree to a tensor involves properties from the symmetric monoidal category of representations.

Before we discuss these processes in more detail we do some set-up by defining formally a ‘tensor species’.

## 2.1. TENSOR SPECIES

A tensor species is a formalization of the data needed to define e.g., complex Lorentz tensors, real Lorentz tensors or Einstein tensors. The word ‘species’ is a nod to ‘graphical species’ defining in [js: ...](#), from which our construction is inspired.

We will start by giving the complete definition of a tensor species in Lean, and then dissect this definition, discussing each of the components in turn.

In Lean a tensor species is defined as follows:

DRAFT

```

/-- The sturcture of a type of tensors e.g. Lorentz tensors, Einstien tensors,
complex Lorentz tensors. -/
structure TensorSpecies where
/-- The commutative ring over which we want to consider the tensors to live in,
usually 'ℝ' or 'ℂ'. -/
k : Type
/-- An instance of 'k' as a commutative ring. -/
k_commRing : CommRing k
/-- The symmetry group acting on these tensor e.g. the Lorentz group or SL(2,ℂ). -/
G : Type
/-- An instance of 'G' as a group. -/
G_group : Group G
/-- The colors of indices e.g. up or down. -/
C : Type
/-- A functor from 'C' to 'Rep k G' giving our building block representations.
Equivalently a function 'C → Re k G'. -/
FDiscrete : Discrete C ⇒ Rep k G
/-- A specification of the dimension of each color in C. This will be used for
explicit
evaluation of tensors. -/
repDim : C → ℕ
/-- repDim is not zero for any color. This allows casting of 'ℕ' to 'Fin (S.repDim
c)'. -/
repDim_neZero (c : C) : NeZero (repDim c)
/-- A basis for each Module, determined by the evaluation map. -/
basis : (c : C) → Basis (Fin (repDim c)) k (FDiscrete.obj (Discrete.mk c)).V
/-- A map from 'C' to 'C'. An involution. -/
τ : C → C
/-- The condition that 'τ' is an involution. -/
τ_involution : Function.Involutive τ
/-- The natural transformation describing contraction. -/
contr : OverColor.Discrete.pairr FDiscrete τ →  $\mathbb{1}_$  (Discrete C ⇒ Rep k G)
/-- Contraction is symmetric with respect to duals. -/
contr_tmul_symm (c : C) (x : FDiscrete.obj (Discrete.mk c))
(y : FDiscrete.obj (Discrete.mk (τ c))) :
(contr.app (Discrete.mk c)).hom (x  $\otimes$ ,[k] y) = (contr.app (Discrete.mk (τ c))).hom
(y  $\otimes$ , (FDiscrete.map (Discrete.eqToHom (τ_involution c)).symm)).hom x)
/-- The natural transformation describing the unit. -/
unit :  $\mathbb{1}_$  (Discrete C ⇒ Rep k G) → OverColor.Discrete.τPair FDiscrete τ
/-- The unit is symmetric. -/
unit_symm (c : C) :
((unit.app (Discrete.mk c)).hom (1 : k)) =
((FDiscrete.obj (Discrete.mk (τ c))))  $\triangleleft$ 
(FDiscrete.map (Discrete.eqToHom (τ_involution c))))).hom
((β_ (FDiscrete.obj (Discrete.mk (τ (τ c)))) (FDiscrete.obj (Discrete.mk (τ
(c)))))).hom.hom
((unit.app (Discrete.mk (τ c))).hom (1 : k)))
/-- Contraction with unit leaves invariant. -/
contr_unit (c : C) (x : FDiscrete.obj (Discrete.mk (c))) :
(λ_ (FDiscrete.obj (Discrete.mk (c))))).hom.hom
(((contr.app (Discrete.mk c))  $\triangleright$  (FDiscrete.obj (Discrete.mk (c))))).hom
((α_ _ _ (FDiscrete.obj (Discrete.mk (c))))).inv.hom
(x  $\otimes$ ,[k] (unit.app (Discrete.mk c)).hom (1 : k))) = x

```

Let us go through this definition piece by piece. Firstly  $k$  is a type, has the instance  $k\_commRing$  of a commutative ring. It represents the ring (or field) that our tensors are defined over. For complex Lorentz tensors, this is naturally, the field of complex numbers.

The type  $G$  has the instance of a group  $G\_group$ . It represents the group which acts on our tensors. For complex Lorentz tensors, this is the group  $SL(2, \mathbb{C})$ , whilst for real Lorentz tensors this is the Lorentz group.

The type  $C$  is a type of colors. For complex Lorentz tensors, there are six colors forming the type

```
inductive Color
| upL : Color
| downL : Color
| upR : Color
| downR : Color
| up : Color
| down : Color
```

Colors can be thought of as labels for each of the building block representations making up the tensor species. This is made formal by the functor  $F_{Discrete}$ , which assigns to each color a representation of  $G$  over  $k$ . In what follows we will use  $F_D$  to denote this functor. For complex Lorentz tensors, this is defined as follows:

```
FDiscrete := Discrete.functor fun c =>
match c with
| Color.upL => Fermion.leftHanded
| Color.downL => Fermion.altLeftHanded
| Color.upR => Fermion.rightHanded
| Color.downR => Fermion.altRightHanded
| Color.up => Lorentz.complexContr
| Color.down => Lorentz.complexCo
```

The representations defined here are:

- The representation of left-handed Weyl fermions, denoted in Lean as `Fermion.leftHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto Mv$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of ‘alternative’ left-handed Weyl fermions, denoted in Lean as `Fermion.altLeftHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto M^{-1T}v$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of right-handed Weyl fermions, denoted in Lean as `Fermion.rightHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto M^*v$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of ‘alternative’ right-handed Weyl fermions, denoted in Lean as `Fermion.altRightHanded`, and corresponding to the representation of  $SL(2, \mathbb{C})$  taking  $v \mapsto M^{-1\dagger}v$  for  $M \in SL(2, \mathbb{C})$ .
- The representation of contravariant Lorentz tensors, denoted in Lean as `Lorentz.complexContr`, and corresponding to the representation of  $SL(2, \mathbb{C})$  induced by the homomorphism of  $SL(2, \mathbb{C})$  into the Lorentz group and the contravariant action of the Lorentz group on four-vectors.

- The representation of covariant Lorentz tensors, denoted in Lean as `Lorentz.complexCo`, and corresponding to the representation of  $SL(2, \mathbb{C})$  induced by the homomorphism of  $SL(2, \mathbb{C})$  into the Lorentz group and the covariant action of the Lorentz group on four-vectors.

As an example, the representation of left-handed Weyl fermions `Fermion.leftHanded` is defined in Lean as follows:

```

/-- The vector space  $\mathbb{C}^2$  carrying the fundamental representation of  $SL(2, \mathbb{C})$ .
    In index notation corresponds to a Weyl fermion with indices  $\psi^a$ . -/
def leftHanded : Rep  $\mathbb{C}$   $SL(2, \mathbb{C})$  := Rep.of {
  /- The function from  $SL(2, \mathbb{C})$  to endomorphisms of LeftHandedModule
      (which corresponds to the vector space  $\mathbb{C}^2$ ). -/
  toFun := fun M => {
    /- Start of the definition of the linear map. -/
    /- The function underlying the linear map. Defined as the dot product. -/
    toFun := fun ( $\psi$  : LeftHandedModule) =>
      LeftHandedModule.toFin2CEquiv.symm (M.1 *v  $\psi$ .toFin2C),
    /- Proof that the function is linear with respect to addition. -/
    map_add' := by
      intro  $\psi$   $\psi'$ 
      simp [mulVec_add]
    /- Proof that the function is linear with respect to scalar multiplication. -/
    map_smul' := by
      intro r  $\psi$ 
      simp [mulVec_smul]
    /- End of the definition of the linear map. -/
  }
  /- Proof that (the outer) toFun gives the identity map on the identity of
       $SL(2, \mathbb{C})$ . -/
  map_one' := by
    ext i
    simp
  /- Proof that the action of the product of two elements is
      the product of the actions of the elements. -/
  map_mul' := fun M N => by
    simp only [SpecialLinearGroup.coe_mul]
    ext1 x
    simp only [LinearMap.coe_mk, AddHom.coe_mk, LinearMap.mul_apply,
      LinearEquiv.apply_symm_apply,
      mulVec_mulVec]}

```

We have added some explanatory comments to this code, not seen in the actual Lean code, to give the reader an idea of what each part does.

The map `repDim` assigns to each color a natural number, which corresponds to the dimension of the representation associated to that color. The condition is placed on the representations that they are non-empty, i.e., that the dimension is not equal to zero `repDim_neZero`. The map `basis` gives a basis indexed by `(Fin (repDim c))` (numbers from 0 to `repDim c - 1`) of each representation for each  $c \in C$ . We will use this basis in the definition of evaluation of tensor indices.

To each color the map  $\tau$ , which is an involution via  `$\tau$ _involution`, defines a dual color. The dual of a color is that which it can be contracted with. So, for complex Lorentz tensors the map  $\tau$  is given by

```

τ := fun c =>
  match c with
  | Color.upL => Color.downL
  | Color.downL => Color.upL
  | Color.upR => Color.downR
  | Color.downR => Color.upR
  | Color.up => Color.down
  | Color.down => Color.up

```

The contraction is defined by `contr`, which is a natural transformation from the functor  $F_D(\_) \otimes F_D(\tau(\_))$ , to the functor  $\mathbb{1}$  takes every object in  $C$  to the trivial representation  $k$ . This natural transformation is simply the assignment to each color  $c$  a map from  $F_D(c) \otimes F_D(\tau(c))$  to  $k$  which is equivariant with respect to the group action. This contraction cannot be defined arbitrarily, but must satisfy the symmetry condition `contr_tmul_symm`, which corresponds to the diagram

$$\begin{array}{ccc}
 F_D(c) \otimes F_D(\tau c) & \xrightarrow{\beta} & F_D(\tau c) \otimes F_D(c) \\
 \mathcal{C}(c) \downarrow & & \downarrow \mathbb{1} \otimes F_D \tau_c \\
 \mathbb{1} & \xleftarrow{\mathcal{C}(\tau(c))} & F_D(\tau c) \otimes F_D(\tau \tau c)
 \end{array} \quad (1)$$

commuting, where  $\beta$  is the braiding of the symmetric monoidal category, and  $\tau_c$  is the isomorphism in  $C$  between  $\tau(\tau(c))$  and  $c$ , which exists since  $\tau$  is an involution.

Along with the contraction we define the `unit`, along with its own symmetry condition `unit_symm`. The unit is a natural transformation from the functor  $\mathbb{1}$  to  $F_D(\tau(\_)) \otimes F_D(\_)$ . This is the assignment to each color  $c$  an object of  $F_D(\tau(c)) \otimes F_D(c)$  which is invariant with respect to the group action. The symmetry condition `unit_symm` is represented by the diagram

$$\begin{array}{ccc}
 \mathbb{1} & \xrightarrow{\delta(c)} & F_D(\tau c) \otimes F_D(c) \\
 \delta(\tau(c)) \downarrow & & \uparrow \mathbb{1} \otimes F_D \tau_c \\
 F_D(\tau \tau c) \otimes F_D(\tau c) & \xrightarrow{\beta} & F_D(\tau c) \otimes F_D(\tau \tau c)
 \end{array} \quad (2)$$

The unit is actually a unit in the sense that contraction with it does nothing. This is made formal with the condition `contr_unit`, which corresponds to the diagram

$$\begin{array}{ccccc}
 F_D(c) & \xrightarrow{\rho^{-1}} & F_D(c) \otimes \mathbb{1} & \xrightarrow{\mathbb{1} \otimes \delta(c)} & F_D(c) \otimes (F_D(\tau c) \otimes F_D(c)) \\
 \lambda \uparrow & & & & \downarrow \alpha^{-1} \\
 \mathbb{1} \otimes F_D(c) & \xleftarrow{\mathcal{C}(c) \otimes \mathbb{1}} & & & (F_D(c) \otimes F_D(\tau c)) \otimes F_D(c)
 \end{array} \quad (3)$$

where  $\rho$  is the right-unitor,  $\lambda$  the left-unitor, and  $\alpha$  is the associator.

The final part of the definition of a tensor species is the definition of the metric, `metric` and its interaction with the contraction and unit `contr_metric`. The metric is a natural transformation from the functor  $\mathbb{1}$  to the functor  $F_D(\_) \otimes F_D(\_)$ . It thus represents the assignement to each color  $c$  an object of  $F_D(c) \otimes F_D(c)$  which is invariant with respect to the group action. The metric can be used to change an

index into a dual index. The condition `contr_metric` corresponds to the diagram

$$\begin{array}{ccc}
 \mathbb{I} & \xrightarrow{\delta(c)} & F_D(\tau c) \otimes F_D(c) \\
 \downarrow \eta(c) \otimes \eta(\tau(c)) & & \uparrow \beta \\
 (F_D c \otimes F_D c) \otimes (F_D(\tau c) \otimes F_D(\tau c)) & & F_D(c) \otimes F_D(\tau c) \\
 \downarrow & & \uparrow 1 \otimes (\mathcal{C}(c) \otimes 1) \\
 F_D c \otimes (F_D c \otimes (F_D(\tau c) \otimes F_D(\tau c))) & \longrightarrow & F_D c \otimes ((F_D c \otimes F_D(\tau c)) \otimes F_D(\tau c))
 \end{array} \tag{4}$$

## 2.2. TENSORS

Given any type  $C$ , which we will think of as a type of colors, we can define the category  $T_{/C}^\times$  as follows. Objects are functions  $f : X \rightarrow C$  for some type  $X$ . A morphism from  $f : X \rightarrow C$  to  $g : Y \rightarrow C$  is a bijection  $\phi : X \rightarrow Y$  such that  $f = g \circ \phi$ . This category is equivalent to the core of the category of types sliced over  $C$ , which explains the origin of the notation. In Lean we denote this category as `OverColor C`, although we will use the shorter notation  $T_{/C}^\times$  here when we can.

The category  $T_{/C}^\times$  is not any old category, it carries a symmetric monoidal structure, which we will denote  $\otimes$ . The structure such that  $f \otimes g$  for objects  $f$  and  $g$  is the map  $X \oplus Y \rightarrow C$  where  $\oplus$  denotes the disjoint union of types.

The functor  $F_D$  defined above, can be lifted to a symmetric monoidal functor  $F$  from  $T_{/C}^\times$  to  $\text{Rep } k G$ . This functor takes  $f : X \rightarrow C$  to the tensor product  $\bigotimes_{x \in X} F_D(f(x))$  and morphisms to the linear maps of representations corresponding to reindexings of tensor products. This construction is general, and functorial. In otherwords, there is a functor

$$\text{lift} : \text{Fun}(C, \text{Rep } k G) \Rightarrow \text{SymmMonFun}(T_{/C}^\times, \text{Rep } k G) \tag{5}$$

In Lean we define this functor, along with the corresponding lift of  $F_D$ ,  $F$ .

We can think of an object  $f : X \rightarrow C$  of  $T_{/C}^\times$  as a type of indices  $X$ , and a specification of what representation each index is associated to. The representation  $F(f)$  is then the tensor product of each of these representations, so that vectors  $v \in F(f)$  can be thought of as tensors with indices indexed by  $X$  of color  $C$  (e.g. ‘up’ or ‘down’).

With this in mind, we define a general tensor of a given species as a vector in the representation  $F(f)$  for some  $f$  in  $T_{/C}^\times$ .

In physics, we typically focus on objects  $f : X \rightarrow C$  in  $T_{/C}^\times$  where  $X$  is a finite type of the form  $\text{Fin } n$  for some  $n$ . Here,  $\text{Fin } n$  represents the type of natural numbers less than  $n$ , i.e.,  $\{0, 1, \dots, n-1\}$ . When appropriate in what follows, we will restrict to these objects.

In Lean for a map of types  $f : X \rightarrow C$ , what we have been writing as  $F(f)$  is written as `S.F.obj (OverColor.mk f)`. The object `S` is the tensor species, which we will define shortly. The `OverColor.mk` tells lean that the function `f` should be considered as an object of `OverColor C`, not just as an object of type  $X \rightarrow C$ . [js: fix arrows here](#)

## 2.3. TENSOR TREES AND THEIR MAP TO TENSORS

Tensor trees are trees with a node for each of the basic operations one can perform on a tensor. Namely, tensor trees have nodes for addition of tensors, permutation of tensor indices, negation of tensors, scalar



multiplication of tensors, group action on a tensor, tensor product of tensors, contraction of tensor indices, and evaluation of tensor indices. They also have nodes for tensors themselves.

Given a species  $S$ , we have a type of tensor tree for each map of the form  $c : \text{Fin } n \rightarrow S.C$  in  $\text{OverColor } S.C$ . This restriction to  $\text{Fin } n$  is done for convenience.

Tensor trees are defined inductively through a number of constructors:

```
inductive TensorTree (S : TensorSpecies) : {n : ℕ} → (Fin n → S.C) → Type where
  /-- A general tensor node. -/
  | tensorNode {n : ℕ} {c : Fin n → S.C} (T : S.F.obj (OverColor.mk c)) :
    TensorTree S c
  /-- A node corresponding to the scalar multiple of a tensor by a element of the
    field. -/
  | smul {n : ℕ} {c : Fin n → S.C} : S.k → TensorTree S c → TensorTree S c
  /-- A node corresponding to negation of a tensor. -/
  | neg {n : ℕ} {c : Fin n → S.C} : TensorTree S c → TensorTree S c
  /-- A node corresponding to the addition of two tensors. -/
  | add {n : ℕ} {c : Fin n → S.C} : TensorTree S c → TensorTree S c → TensorTree
    S c
  /-- A node corresponding to the action of a group element on a tensor. -/
  | action {n : ℕ} {c : Fin n → S.C} : S.G → TensorTree S c → TensorTree S c
  /-- A node corresponding to the permutation of indices of a tensor. -/
  | perm {n m : ℕ} {c : Fin n → S.C} {c1 : Fin m → S.C}
    (σ : (OverColor.mk c) → (OverColor.mk c1)) (t : TensorTree S c) :
    TensorTree S c1
  /-- A node corresponding to the product of two tensors. -/
  | prod {n m : ℕ} {c : Fin n → S.C} {c1 : Fin m → S.C}
    (t : TensorTree S c) (t1 : TensorTree S c1) : TensorTree S (Sum.elim c c1 o
    finSumFinEquiv.symm)
  /-- A node corresponding to the contraction of indices of a tensor. -/
  | contr {n : ℕ} {c : Fin n.succ.succ → S.C} : (i : Fin n.succ.succ) →
    (j : Fin n.succ) → (h : c (i.succAbove j) = S.τ (c i)) → TensorTree S c →
    TensorTree S (c o Fin.succAbove i o Fin.succAbove j)
  /-- A node corresponding to the evaluation of an index of a tensor. -/
  | eval {n : ℕ} {c : Fin n.succ → S.C} : (i : Fin n.succ) → (x : ℕ) →
    TensorTree S c →
    TensorTree S (c o Fin.succAbove i)
```

Each constructor here, e.g. `tensorNode`, `smul`, `neg`, etc., can be thought of as forming a different type of node in a tensor tree.

Since the interpretation of each of the constructors is down to how we turn them into a tensor, we discuss this before discussing each of the constructors in turn. The process of going from a tensor tree to a tensor is proscribed by a function  $\text{TensorTree } S \text{ } c \rightarrow S.F.\text{obj } (\text{OverColor.mk } c)$ , which is defined recursively as follows:

```

/-- The underlying tensor a tensor tree corresponds to. -/
def tensor {n : ℕ} {c : Fin n → S.C} : TensorTree S c → S.F.obj (OverColor.mk c)
  := fun
  | tensorNode t => t
  | smul a t => a · t.tensor
  | neg t => - t.tensor
  | add t1 t2 => t1.tensor + t2.tensor
  | action g t => (S.F.obj (OverColor.mk _)).ρ g t.tensor
  | perm σ t => (S.F.map σ).hom t.tensor
  | prod t1 t2 => (S.F.map (OverColor.equivToIso finSumFinEquiv).hom).hom
    ((S.F.μ _ _).hom (t1.tensor ⊗t t2.tensor))
  | contr i j h t => (S.contrMap _ i j h).hom t.tensor
  | eval i e t => (S.evalMap i (Fin.ofNat' _ e)) t.tensor

```


Let us now dicuss each of the constructors in turn, after which we will give some examples tensor trees.

**tensorNode:** The constructor `tensorNode` based on `c` creates a tensor tree from a tensor `t` in `S.F.obj (OverColor.mk c)`. This tensor tree consists of a single node that directly represents the tensor. Since all other tensor tree constructors require an existing tensor tree as input, `tensorNode` serves as the foundational base case for building more complex trees. Diagrammatically, such a tree is




Naturally, the tensor associated with this node is exactly the tensor provided during construction.

**smul:** The constructor `smul` takes a scalar `a` and an existing tensor tree `t` based on `c` and constructs a new tensor tree also based on `c`. Conceptually, this new tree has a root node labeled `smul a`, with the tensor tree `t` as its child. Diagrammatically, such a tree is




The tensor associated with this new tree is obtained by multiplying the tensor associated with `t` by the scalar `a`.

**neg:** The constructor `neg` takes an existing tensor tree `t` based on `c` and constructs a new tensor tree also based on `c`. This new tree has a root node labeled `neg`, with the tensor tree `t` as its child. Diagrammatically, we have




The tensor associated with this new tree is obtained by negating the tensor associated with `t`.

**add:** The constructor `add` takes two existing tensor trees `t1` and `t2`, based on the same `c`, and constructs a new tensor tree. This new tree has a root node labeled `add`, with the tensor trees `t1` and `t2` as its children. Diagrammatically, this corresponds to




The tensor associated with this new tree is obtained by adding the tensors associated with  $t1$  and  $t2$ .

**action:** The constructor `action` takes a group element  $g$  of  $S.G$ , an existing tensor tree  $t$  based on  $c$ , and constructs a new tensor tree also based on  $c$ . This new tree has a root node labeled `action g`, with the tensor tree  $t$  as its child. Diagrammatically,




The tensor associated with this new tree is obtained by acting on the tensor associated with  $t$  with the group element  $g$ .

**perm:** The constructor `perm` takes a morphism  $\sigma$  from  $c$  to  $c1$  in  $\text{OverColor } S.C$  and an existing tensor tree  $t$  based on  $c$ , and constructs a new tensor tree based on  $c1$ . This new tree has a root node labeled `perm  $\sigma$` , with the tensor tree  $t$  as its child. Diagrammatically, this is given by



The tensor associated with this new tree is obtained by applying the image of the morphism  $\sigma$  under the functor  $S.F$  to the tensor associated with  $t$ .


**prod:** The constructor `prod` takes two existing tensor trees  $t$  based on  $c$  and  $t1$  based on  $c1$  and constructs a new tensor tree based on  $\text{Sum.elim } c \ c1 \circ \text{finSumFinEquiv.symm}$  which is the map from  $\text{Fin } (n + n1)$  acting via  $c$  on  $i = 0, \dots, n-1$  and via  $c1$  on  $i = n, \dots, n+n1-1$ . This new tree has a root node labeled `prod`, with the tensor trees  $t$  and  $t1$  as its children. Diagrammatically, this is given by



The tensor associated with this new tree is obtained by taking the tensor product of the tensors associated with  $t$  and  $t1$ , giving a vector in  $S.F.\text{obj } (\text{OverColor.mk } c) \otimes S.F.\text{obj } (\text{OverColor.mk } c1)$ , using the tensorate of  $S.F$  to map this vector into a vector in  $S.F.\text{obj } (\text{OverColor.mk } c \otimes \text{OverColor.mk } c1)$ , and finally using an isomorphism between  $\text{OverColor.mk } c \otimes \text{OverColor.mk } c1$  and  $\text{OverColor.mk } (\text{Sum.elim } c \ c1 \circ \text{finSumFinEquiv.symm})$  to map this vector into  $S.F.\text{obj } (\text{OverColor.mk } (\text{Sum.elim } c \ c1 \circ \text{finSumFinEquiv.symm}))$ .


**contr:** The constructor `contr` firstly, takes an existing tensor tree  $t$  based on a  $c : \text{Fin } n.\text{succ.succ} \rightarrow S.C$ . Here  $n.\text{succ.succ}$  is  $n+1+1$  with `succ` meaning the successor of a natural number. It also takes, an  $i$  of type  $\text{Fin } n.\text{succ.succ}$ ,  $j$  of type  $\text{Fin } n.\text{succ}$  and a proof that  $c \ (i.\text{succAbove } j) = S.\tau \ (c \ i)$ , where  $i.\text{succAbove}$  is the map from  $\text{Fin } n.\text{succ}$  to  $\text{Fin } n.\text{succ.succ}$  with a hole at  $i$ . The proof says that the color of the index  $i.\text{succAbove } j$  is the dual of the color of the index  $i$ , and thus that these two indices can be contracted. Note that we use a  $j$  in  $\text{Fin } n.\text{succ}$  and  $i.\text{succAbove } j$  as the index to be contracted, instead of another index in  $\text{Fin } n.\text{succ.succ}$  to ensure the two indices to be contracted

are not the same. The constructor outputs a new tensor tree based on  $c \circ i.\text{succAbove} \circ j.\text{succAbove}$ . This new tree has a root node labeled  $\text{contr } i \ j$ , with the tensor tree  $t$  as its child. Diagrammatically,



The tensor associated with the new tensor tree is constructed as follows. We start with a tensor in  $S.F.\text{obj } (\text{OverColor.mk } c)$ . This is mapped into a vector in  $(S.F\text{Discrete.obj } (c \ i)) \otimes S.F\text{Discrete.obj } (S.\tau (c \ i))) \otimes S.F.\text{obj } (\text{OverColor.mk } (c \circ \text{Fin.succAbove } i \circ \text{Fin.succAbove } j))$  via an equivalence from  $S.F.\text{obj } (\text{OverColor.mk } c)$ . The equivalence is constructed using an equivalence in  $\text{OverColor } C$  to extract  $i$  and  $i.\text{succAbove } j$  from  $c$ , then using the tensorate of  $S.F.$ , and the fact that  $c \ (i.\text{succAbove } j) = S.\tau (c \ i)$ . Using the contraction for  $c \ i$ , we then get a vector in  $\mathbb{1} \otimes S.F.\text{obj } (\text{OverColor.mk } (c \circ \text{Fin.succAbove } i \circ \text{Fin.succAbove } j))$  which is then mapped to a tensor in  $S.F.\text{obj } (\text{OverColor.mk } (c \circ \text{Fin.succAbove } i \circ \text{Fin.succAbove } j))$  using the left-unitor of  $\text{Rep } S.G \ S.k$ . **is: Relate this back to the tensor construction.**

**eval:** The final constructor **eval** takes an existing tensor tree  $t$  based on a  $c : \text{Fin } n.\text{succ} \rightarrow S.C$ , an  $i$  of type  $\text{Fin } n.\text{succ}$ , and a natural number  $x$ . The constructor outputs a new tensor tree based on  $c \circ \text{Fin.succAbove } i$ . This new tree has a root node labeled **eval**  $i \ x$ , with the tensor tree  $t$  as its child. Diagrammatically, this corresponds to



The tensor associated with the new tensor tree is constructed as follows. We start with a tensor in  $S.F.\text{obj } (\text{OverColor.mk } c)$ . This is mapped into a vector in  $S.F\text{Discrete.obj } (c \ i) \otimes S.F.\text{obj } (\text{OverColor.mk } (c \circ \text{Fin.succAbove } i))$  via an equivalence from  $S.F.\text{obj } (\text{OverColor.mk } c)$ . The equivalence is constructed using an equivalence in  $\text{OverColor } C$  to extract  $i$  from  $c$ , and then using the tensorate of  $S.F.$ . Using the evaluation for  $c \ i$  at the basis element indicated by  $x$  (if  $x$  is too big a natural number for the number of basis elements it defaults to 0), we then get a vector in  $\mathbb{1} \otimes S.F.\text{obj } (\text{OverColor.mk } (c \circ \text{Fin.succAbove } i))$ . This mapping should be thought of as occurring in the category of modules over  $S.k$ , rather than in  $\text{Rep } S.G \ S.k$ , since the evaluation is not invariant under the group action. We then finally map into  $S.F.\text{obj } (\text{OverColor.mk } (c \circ \text{Fin.succAbove } i))$  using the left-unitor.

The main reason tensor trees are nice to work with is that they have the following property: Define a sub-tree of a tensor trees to be a node and all child nodes of that node. If  $T$  is a tensor tree and  $S$  a sub-tree of  $T$ , we can replace  $S$  in  $T$  with another tensor tree  $S'$  to get a new overall tensor-tree  $T'$ . If  $S$  and  $S'$  have the same underlying tensor, then  $T$  and  $T'$  will.

In Lean this property manifests in a series of lemmas. For instance, for instance for the **contr** constructor we have the lemma:

```

lemma contr_tensor_eq {n : ℕ} {c : Fin n.succ.succ → S.C} {T1 T2 : TensorTree S c}
  (h : T1.tensor = T2.tensor) {i : Fin n.succ.succ} {j : Fin n.succ}
  {h' : c (i.succAbove j) = S.τ (c i)} :
  (contr i j h' T1).tensor = (contr i j h' T2).tensor := by
simp only [Nat.succ_eq_add_one, contr_tensor]
rw [h]

```

These lemmas allow us to navigate to the certain places in tensor trees and replace sub-trees with other sub-trees. We will see this used extensively in the examples.

## 2.4. SYNTAX AND THEIR MAP TO TENSOR TREES

We now discuss how we make the Lean code look similar to what we would use on pen-and-paper physics.

This is done using a two step process. Firstly, we define syntax for tensor expressions. Then we write code to turn this syntax into a tensor tree. This process is not formally defined or verified, Lean takes the outputted tensor tree as the formal object to work with.

Instead of delving into the nitty-gritty details of how this process works under the hood, we give some examples to see how it works.

In what follows we will assume that  $T$ , and  $T_i$  etc are tensors defined as  $S.F.obj\_$  for some tensor species  $S$ . We will also assume that these tensors are defined correctly for the expressions below to make sense.

The syntax allows us to write the following

$\{T \mid \mu \ v\}^T$	tensorNode T
------------------------	--------------

for a tensor node. Here the  $\mu$  and  $v$  are free variables and it does not matter what we call them - Lean will elaborate the expression in the same way. The elaborator also knows how many indices to expect for a tensor  $T$  and will raise an error if the wrong number are given. The  $\{\_ \}^T$  notation is used to tell Lean that the syntax is to be treated as a tensor expression.

We can write e.g.

$\{T \mid \mu \ v\}^T.tensor$	(tensorNode T).tensor
-------------------------------	-----------------------

to get the underlying tensor.

Note that we have not lowered or risen the indices, as one would expect from pen-and-paper notation. There is one primary reason for this, whether an index is lower or risen does not carry any information, since this information comes from the tensor itself. Also, for something like complex Lorentz tensors, there are at least three different types of upper-index.

We can extract the tensor from a tensor tree using the following syntax. If we want to evaluate an index we can put an explicit index in place of  $\mu$  or  $v$  above, for example

$\{T \mid 1 \ v\}^T$	eval 0 1 (tensorNode T)
----------------------	-------------------------

The syntax and elaboration for negation, scalar multiplication and the group action are fairly similar. For negation we have

$\{T \mid \mu \ v\}^T$	neg (tensorNode T)
------------------------	--------------------

For scalar multiplication by  $a \in k$  we have

$$\{a \cdot T \mid \mu \nu\}^T \quad \text{smul } a \text{ (tensorNode } T)$$

For the group action of  $g \in G$  on a tensor  $T$  we have

$$\{g \cdot_a T \mid \mu \nu\}^T \quad \text{action } g \text{ (tensorNode } T)$$

For the product of two tensors is also fairly simple, we have

$$\{T \mid \mu \nu \otimes T2 \mid \sigma\}^T \quad \text{prod (tensorNode } T) \text{ (tensorNode } T2)$$

The syntax for contraction is as one expect,

$$\{T \mid \mu \nu \otimes T2 \mid \nu \sigma\}^T \quad \text{contr } 1 \ 1 \text{ rfl (prod (tensorNode } T) \text{ (tensorNode } T2))}$$

On the RHS here the first argument (1) of `contr` is the index of the first  $\nu$  on the LHS, the second argument (also 1) is the second index. The `rfl` is a proof that the colors of the two contracted indices are actually dual to one another. If they are not, this proof will fail and the elaborator will complain. It will also complain if more then two indices are trying to be contracted. Although, this depends on where exactly the indices sit in the expression. For example

$$\{T \mid \mu \nu \otimes T2 \mid \nu \nu\}^T \quad (\text{prod (tensorNode } T) \text{ (contr } 0 \ 0 \text{ rfl (tensorNode } T2)))$$

works fine because the inner contraction is done before the product.

We now turn to addition. Our syntax allows for e.g.  $\{T \mid \mu \nu + T2 \mid \mu \nu\}^T$  and also  $\{T \mid \mu \nu + T2 \mid \nu \mu\}^T$ , provided of course that the indices are of the correct color (which Lean will check). The elaborator handles both these cases, and generalizations thereof by adding a permutation node. Thus we have

$$\{T \mid \mu \nu + T2 \mid \mu \nu\}^T \quad \text{add (tensorNode } T) \text{ (perm } \_ \text{ (tensorNode } T2))}$$

where here the  $\_$  is a placeholder for the permutation, and in this case will be trivial, but for

$$\{T \mid \mu \nu + T2 \mid \nu \mu\}^T \quad \text{add (tensorNode } T) \text{ (perm } \_ \text{ (tensorNode } T2))}$$

it will be the permutation for the two identities.

Despite not forming part of a node in our tensor tree, we also give syntax for equality. This is done in a very similar way to addition, with the addition of a permutation node to account for e.g. the fact that  $T_{\mu\nu} = T_{\nu\mu}$ .

$$\{T \mid \mu \nu = T2 \mid \nu \mu\}^T \quad (\text{tensorNode } T). \text{tensor} = (\text{perm } \_ \text{ (tensorNode } T2)). \text{tensor}$$

Note the use of the `.tensor` to extract the tensor from the tensor tree, it does not really mean much to ask for equality of the tensor trees themselves.

### 3. EXAMPLES

We give two examples in this section. The first example is a simple theorem involving index notation and tensor trees. We will show here, in rather explicit detail, how we can manipulate tensor trees to solve such theorems. The second example we shall give will show a number of definitions in HepLean concerning index notation. Here we will not give so much detail, the point rather being to show the reader the broad use of our construction.

### 3.1. EXAMPLE 1: SYMMETRIC AND ANTI-SYMMETRIC TENSOR

If  $A^{\mu\nu}$  is an anti-symmetric tensor and  $S_{\mu\nu}$  and  $S$  is a symmetric tensor, then it is true that  $A^{\mu\nu}S_{\mu\nu} = -A^{\mu\nu}S_{\mu\nu}$ . In Lean this result, and it's lemma are written as follows:

```
lemma antiSymm_contr_symm
  {A : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.up])}
  {S : complexLorentzTensor.F.obj (OverColor.mk ![Color.down, Color.down])}
  (hA : {A |  $\mu \nu = - (A | \nu \mu)^T$ }) (hs : {S |  $\mu \nu = S | \nu \mu^T$ }) :
  {A |  $\mu \nu \otimes S | \mu \nu = - A | \mu \nu \otimes S | \mu \nu^T$ } := by
conv =>
  lhs
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_tensor_eq_fst <| hA]
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_tensor_eq_snd <| hs]
  rw [contr_tensor_eq <| contr_tensor_eq <| prod_perm_left _ _ _]
  rw [contr_tensor_eq <| contr_tensor_eq <| perm_tensor_eq <| prod_perm_right _ _
    _]
  rw [contr_tensor_eq <| contr_tensor_eq <| perm_perm _ _ _]
  rw [contr_tensor_eq <| perm_contr_congr 1 2]
  rw [perm_contr_congr 0 0]
  rw [perm_tensor_eq <| contr_contr _ _ _]
  rw [perm_perm]
  rw [perm_tensor_eq <| contr_tensor_eq <| contr_tensor_eq <| neg_fst_prod _ _]
  rw [perm_tensor_eq <| contr_tensor_eq <| neg_contr _]
  rw [perm_tensor_eq <| neg_contr _]
  apply perm_congr _ rfl
  decide
```

Let us break this down. The statements


```
{A : complexLorentzTensor.F.obj (OverColor.mk ![Color.up, Color.up])}
{S : complexLorentzTensor.F.obj (OverColor.mk ![Color.down, Color.down])}
```

are simply defining  $A$  and  $S$  to be tensors of type  $A^{\mu\nu}$  and  $S_{\mu\nu}$  respectively. This agrees with the notation set out in §?? [js: ref.](#)

The parameter  $hA$  is stating that  $A$  is anti-symmetric. Expanded in terms of tree diagrams we have

$hA : \{A \mid \mu \nu = - (A \mid \nu \mu)^T\}$


Description: The tensor  $A$  is anti-symmetric.




Similarly, the parameter  $hs$  is stating that  $S$  is symmetric. Expanded in terms of tree diagrams

$hS : \{S \mid \mu \nu = S \mid \nu \mu^T\}$

Description: The tensor  $S$  is symmetric.



The line  $\{A \mid \mu v \otimes S \mid \mu v = - A \mid \mu v \otimes S \mid \mu v\}^T$  is the statement we are trying to prove. In terms of tree diagrams it says that




The perm here actually does nothing, but is included by Lean.

The lines of the proof in the `conv` block are manipulations of the tensor tree on the LHS of the equation. The `rw` tactic is used to rewrite the tensor tree using the various lemmas. We go through each step in turn.


```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_tensor_eq_fst <| hA]
```

Description: Replace the node A with the RHS of hA.



```
rw [contr_tensor_eq <| contr_tensor_eq
<| prod_tensor_eq_snd <| hS]
```

Description: Replace the node S with the RHS of hS.






```
rw [contr_tensor_eq <| contr_tensor_eq  
<| prod_perm_left _ _ _]
```

---


Description: Move the left permutation through the product.



```
rw [contr_tensor_eq <| contr_tensor_eq  
<| perm_tensor_eq <| prod_perm_right _ _  
_ _]
```

---


Description: Move the right permutation through the product.



```
rw [contr_tensor_eq <| contr_tensor_eq  
<| perm_perm _ _ _]
```


---

Description: Combine the two permutations (using functoriality).




```
rw [contr_tensor_eq <| perm_contr_congr
1 2]
```

Description: Move the permutation through the contraction. And simplify the contraction indices to 1 and 2 (Lean will check if this is correct).




```
rw [perm_contr_congr 0 0]
```

Description: Move the permutation through the contraction. And simplify the contraction indices to 0 and 0 (Lean will check if this is correct).




```
rw [perm_tensor_eq <| contr_contr _ _ _]
```

Description: Swap the two contractions. This introduces a permutation.




```
rw [perm_perm]
```

Description: Combine the two permutations.




```
rw [perm_tensor_eq <| contr_tensor_eq <|
  contr_tensor_eq <| neg_fst_prod _ _]
```

Description: Move the negation through the product.




```
rw [perm_tensor_eq <| contr_tensor_eq <|
  neg_contr _]
```

Description: Move the negation through the first contraction.



```
rw [perm_tensor_eq <| neg_contr _]
```

Description: Move the negation through the second contraction.



The remainder of the proof `apply perm_congr _ rfl` and decide that the tensor trees on the LHS and RHS are actually equal.

### 3.2. EXAMPLE 2: PAULI MATRICES AND BISPINORS

Using the formlism we have set up thus far it is possible to define Pauli matrices and bispinors as complex Lorentz tensors.

The pauli matrices appear in HepLean as follows

```
/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma^{\mu\alpha\dot{\beta}}$ '. -/
def pauliContr := {PauliMatrix.asConsTensor |  $v^\alpha \beta$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma_{\mu\alpha\dot{\beta}}$ '. -/
def pauliCo := { $\eta' | \mu^\nu \otimes \text{pauliContr} | v^\alpha \beta$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma_{\mu\alpha\dot{\beta}}$ '. -/
def pauliCoDown := {pauliCo |  $\mu^\alpha \beta \otimes \varepsilon_L' | \alpha \alpha' \otimes \varepsilon_R' | \beta \beta'$ }T.tensor

/-- The Pauli matrices as the complex Lorentz tensor ' $\sigma^{\mu\alpha\dot{\beta}}$ '. -/
def pauliContrDown := {pauliContr |  $\mu^\alpha \beta \otimes \varepsilon_L' | \alpha \alpha' \otimes \varepsilon_R' | \beta \beta'$ }T.tensor
```

The first of these definitions depends on `PauliMatrix.asConsTensor` which is defined as using an explicit basis expansion.

In these expressions we have the appearance of metrics  $\eta'$  is the metric usually written as  $\eta_{\mu\nu}$ , `js` etc.

With these we can define bispinors

```
/-- A bispinor ' $p^{aa}$ ' created from a lorentz vector ' $p^\mu$ '. -/
def contrBispinorUp (p : complexContr) :=
  {pauliCo |  $\mu^\alpha \beta \otimes p | \mu$ }T.tensor

/-- A bispinor ' $p_{aa}$ ' created from a lorentz vector ' $p^\mu$ '. -/
def contrBispinorDown (p : complexContr) :=
  { $\varepsilon_L' | \alpha \alpha' \otimes \varepsilon_R' | \beta \beta' \otimes \text{contrBispinorUp } p | \alpha \beta$ }T.tensor

/-- A bispinor ' $p^{aa}$ ' created from a lorentz vector ' $p_\mu$ '. -/
def coBispinorUp (p : complexCo) := {pauliContr |  $\mu^\alpha \beta \otimes p | \mu$ }T.tensor

/-- A bispinor ' $p_{aa}$ ' created from a lorentz vector ' $p_\mu$ '. -/
def coBispinorDown (p : complexCo) :=
  { $\varepsilon_L' | \alpha \alpha' \otimes \varepsilon_R' | \beta \beta' \otimes \text{coBispinorUp } p | \alpha \beta$ }T.tensor
```

Here `complexContr` and `complexCo` are complex contravariant and covariant Lorentz vectors. Lean knows to treat these as tensors when they appear in tensor expressions.

Using these definitions we can start to prove results about the pauli matrices and bispinors. These proofs rely on essentially the sorts of manipulations in the last section, although in some cases we expand tensors in terms of a basis and use rules about how the basis interacts with the operations in a tensor tree.

Examples of things we have proven range

```
lemma coBispinorDown_eq_pauliContrDown_contr (p : complexCo) :
  {coBispinorDown p |  $\alpha \beta = \text{pauliContrDown} \mid \mu \alpha \beta \otimes p \mid \mu\}^T := \text{by}$ 
```

the proof of which is an application of associativity of the tensor product, and appropriately shuffling around of the contractions.

To more complicated results such as

```
-- The statement that ' $\eta_{\{\mu\nu\}} \sigma^{\{\mu \alpha \text{ dot } \beta\}} \sigma^{\{\nu \alpha' \text{ dot } \beta'\}} = 2 \epsilon^{\{\alpha\alpha'\}} \epsilon^{\{\text{dot } \beta \text{ dot } \beta'\}}$ '. -/
theorem pauliCo_contr_pauliContr :
  {pauliCo |  $\nu \alpha \beta \otimes \text{pauliContr} \mid \nu \alpha' \beta' = 2 \epsilon_L \mid \alpha \alpha' \otimes \epsilon_R \mid \beta \beta'\}^T := \text{by}$ 
```

#### 4. FUTURE WORK

The scale of formalizing all results regarding index notation is a task that surpasses the capacity of any single individual. Inspired by the Lean community's blueprint projects, we have added to HepLean informal lemmas related to index notation and tensors. An example of such is

```
informal_lemma coBispinorUp_eq_metric_contr_coBispinorDown where
  math :≈ "{coBispinorUp p |  $\alpha \beta = \epsilon_L \mid \alpha \alpha' \otimes \epsilon_R \mid \beta \beta' \otimes \text{coBispinorDown p} \mid \alpha, \beta' \}^T"$ 
  proof :≈ "Expand 'coBispinorDown' and use fact that metrics contract to the identity."
  deps :≈ ["coBispinorUp", "coBispinorDown", "leftMetric", "rightMetric"]
```

The lemmas resource that we hope others—or even automated systems—will formalize in the future.

As demonstrated in our earlier examples, manipulating tensor expressions can involve tedious calculations, especially when dealing with directly with tensor trees. In the future, we intend to automate many of these routine steps by developing suitable tactics within Lean. We are optimistic that the structured nature of tensor trees will lend itself well to such automation, thereby streamlining computations and enhancing the efficiency of formal proofs involving index notation and tensor species.

There are two primary directions in which we can extend the concepts presented in this work. First, we could incorporate the spinor-helicity formalism, which is used in the study of scattering amplitudes. Second, we could extend our approach to encompass tensor *fields*, their derivatives etc. We do not anticipate any insurmountable challenges in pursuing these extensions. They represent promising avenues for future research and have the potential to significantly enhance the utility of formal methods in physics.

#### REFERENCES

- [1] Joseph Tooby-Smith. HepLean: Digitalising high energy physics. 5 2024. [arXiv:2405.08863](https://arxiv.org/abs/2405.08863).
- [2] The mathlib Community. The lean mathematical library. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs*, CPP 2020, pages 367–381, New York, NY, USA, 2020. Association for Computing Machinery. [doi:10.1145/3372885.3373824](https://doi.org/10.1145/3372885.3373824).
- [3] Kevin Buzzard and Richard Taylor. A lean proof of fermat's last theorem. 2024.

- [4] Maxwell P Bobbin, Samiha Sharlin, Parivash Feyzishendi, An Hong Dang, Catherine M Wraback, and Tyler R Josephson. Formalizing chemical physics using the lean theorem prover. *Digital Discovery*, 3(2):264–280, 2024. doi:/10.1039/D3DD00077J.
- [5] Tomáš Skřivan. Scilean: Scientific computing assistant. *GitHub repository*. URL: <https://github.com/lecopivo/SciLean>.
- [6] Jean-Philippe Bernardy and Patrik Jansson. Domain-specific tensor languages. *arXiv preprint arXiv:2312.02664*, 2023.

DRAFT